

GAMS 示例

下面这个问题主要是用来举例说明 GAMS 是如何让您以一种自然的方式建立模型.GAMS 能够处理大的多和高度复杂的问题.在这里只展示了 GAMS 的很少一部分的基本特征.

代数描述

下面是问题的标准的代数描述,这是用来最小化把货物从 2 个工厂运输到 3 个市场的费用.受约束于供需约束.

指标:

i =工厂(plants)

j =市场(markets)

给定的数据:

a_i = 在工厂 i 日常供应量(批数 cases)

b_j = 在市场 j 的日常需求量(批数 cases)

d_{ij} = 在工厂 i 和市场 j 之间的距离(千英里)

c_{ij} = 在工厂 i 和市场 j 之间每单位的运输费用(\$/批/千英里)

		距离			
		市场			
工厂	New York	Chicago	Topeka	供应量	
Seattle	2.5	1.7	1.8	350	
San Diego	2.5	1.8	1.4	600	
需求量	325	300	275		

F =每批每千英里的费用\$

决策变量:

x_{ij} = 日常从工厂 i 运输到市场 j 的总量(批数 cases)

这里 $x_{ij} \geq 0$ 适用所有 i, j

约束:

在工厂 i 的供应量限制(批数 cases): $\sum_j x_{i,j} \leq a_i$ 适用所有的 i

在市场 j 的需求量: $\sum_i x_{i,j} \geq b_j$ 适用所有的 j

目标函数:

Minimize $\sum_i \sum_j c_{i,j} x_{i,j}$ (千元)

GAMS 模型

同样的模型在 GAMS 中建模. 简练的代数描述使得模型高度紧凑, 并带有逻辑结构. 内部的文档, 比如对参数的解释和测量的单位, 使得模型很容易读懂.

```
SETS
    I   canning plants   / SEATTLE, SAN-DIEGO /
    J   markets          / NEW-YORK, CHICAGO, TOPEKA / ;
PARAMETERS
    A(I)  capacity of plant i in cases
          /   SEATTLE      350
            SAN-DIEGO     600 /
    B(J)  demand at market j in cases
          /   NEW-YORK     325
            CHICAGO       300
            TOPEKA        275 / ;
TABLE D(I,J)  distance in thousands of miles
           NEW-YORK    CHICAGO    TOPEKA
SEATTLE   2.5         1.7         1.8
SAN-DIEGO 2.5         1.8         1.4 ;
SCALAR F  freight in dollars per case per thousand miles /90/ ;
PARAMETER C(I,J)  transport cost in thousands of dollars per case ;
              C(I,J) = F * D(I,J) / 1000 ;
VARIABLES
    X(I,J)  shipment quantities in cases
    Z       total transportation costs in thousands of dollars ;
POSITIVE VARIABLE X ;
EQUATIONS
    COST          define objective function
    SUPPLY(I)     observe supply limit at plant i
    DEMAND(J)     satisfy demand at market j ;
COST ..          Z =E= SUM((I,J), C(I,J)*X(I,J)) ;
SUPPLY(I) ..    SUM(J, X(I,J)) =L= A(I) ;
DEMAND(J) ..    SUM(I, X(I,J)) =G= B(J) ;
MODEL TRANSPORT /ALL/ ;
SOLVE TRANSPORT USING LP MINIMIZING Z ;
```

集合(Sets)

```
SETS
  I  canning plants   / SEATTLE, SAN-DIEGO /
  J  markets          / NEW-YORK, CHICAGO, TOPEKA / ;
```

GAMS 让您以直接的方式指定指标:声明和命名集合(这里是 I 和 J),并列举它们的元素.

参数

```
PARAMETERS
  A(I)  capacity of plant i in cases
        /   SEATTLE      350
          SAN-DIEGO     600 /
  B(J)  demand at market j in cases
        /   NEW-YORK     325
          CHICAGO       300
          TOPEKA       275 / ;
```

这里的数据输入被作为指标参数 A(I)和 B(J),值简单的被列出.

GAMS 让您可以在模型的任意位置放置解释性文本(以小写格式显示),当您在开发它时,您的注释自动被结合到输出报告中的合适位置.

表格

```
TABLE D(I,J)  distance in thousands of miles
              NEW-YORK      CHICAGO      TOPEKA
  SEATTLE      2.5          1.7          1.8
  SAN-DIEGO    2.5          1.8          1.4 ;
```

数据同样能够以方便的表格形式输入.GAMS 让您以数据的基本形式来输入数据-转换是特定的代数化的.

标量(Scalar)

```
SCALAR F  freight in dollars per case per thousand miles /90/ ;
```

常量能够被声明为标量,它的值是指定的.

数据处理

```
PARAMETER C(I,J)  transport cost in thousands of dollars per case ;
  C(I,J) = F * D(I,J) / 1000 ;
```

当数据值要被计算前,您首先要声明参数(比如,给它一个符号,随意给它编个指标),然后给它一个代数公式.GAMS 将自动进行计算.

变量

```
VARIABLES
    X(I,J)  shipment quantities in cases
    Z       total transportation costs in thousands of dollars ;
    POSITIVE VARIABLE X ;
```

决策变量以代数的方式表达,带有特定的指标.从这种常见的形式,GAMS 在域中生成变量的每个实例.

变量可以被指定为下列类型:任意(FREE),正值(POSITIVE),负值(NEGATIVE),二元(BINARY),或者整数(INTEGER).默认是任意(FREE).

目标变量(这里是 z)仅被声明,没有指标(index).

方程式

```
EQUATIONS
    COST          define objective function
    SUPPLY(I)     observe supply limit at plant i
    DEMAND(J)     satisfy demand at market j ;
    COST ..      Z =E= SUM((I,J), C(I,J)*X(I,J)) ;
    SUPPLY(I) .. SUM(J, X(I,J)) =L= A(I) ;
    DEMAND(J) .. SUM(I, X(I,J)) =G= B(J) ;
```

目标函数和约束方程式首先被通过指定名字来声明.然后它们的概括的算术公式被声明.GAMS 现在已经有了足够的信息(从上面的数据输入和从在方程式中指定的算术关系)来自动生成每个单独的约束声明-就像您能在下面的输出报告中看到的.

=E=表示'equal to' (等于)

=L=表示 'less than or equal to' (小于或等于)

=G=表示'greater than or equal to' (大于或等于)

模型声明

```
MODEL TRANSPORT /ALL/ ;
```

模型被指定了一个唯一的名字(这里是 TRANSPORT),模型缔造者指定那个方程式应该被包含到这个特别的公式中.在这里我们指定了 ALL,也就是说所有的方程式都是模型的一部分.就等于是 MODEL TRANSPORT /COST, SUPPLY, DEMAND/. 这个方程式选择使您能够在单个的 GAMS 输入文件中以公式表达不同的模型,基于相同或不同的给定数据.

求解声明

```
SOLVE TRANSPORT USING LP MINIMIZING Z ;
```

求解声明(1)告诉 GAMS 那个模型要被求解,(2)选择要使用的求解器(在这里是 LP 求解器),(3)表明优化方向,或者是求最小值,或者是求最大值.(4)指定目标变量.

GAMS 输出报告(部分摘录)

完整的 GAMS 输出报告比下面列出的部分摘录详细的多,包含了更多的帮助用于解释和诊断您的模型.甚至您能够修改输出格式来符合您的特定的需要.

方程式列表

```
---- COST          =E=  define objective function

COST..  - 0.225*X(SEATTLE,NEW-YORK) - 0.153*X(SEATTLE,CHICAGO)
        - 0.162*X(SEATTLE,TOPEKA) - 0.225*X(SAN-DIEGO,NEW-YORK)
        - 0.162*X(SAN-DIEGO,CHICAGO) - 0.126*X(SAN-DIEGO,TOPEKA) + Z =E= 0 ;
        (LHS = 0)

---- SUPPLY        =L=  observe supply limit at plant i

SUPPLY(SEATTLE).. X(SEATTLE,NEW-YORK) + X(SEATTLE,CHICAGO)
                 + X(SEATTLE,TOPEKA) =L= 350 ; (LHS = 0)

SUPPLY(SAN-DIEGO).. X(SAN-DIEGO,NEW-YORK) + X(SAN-DIEGO,CHICAGO)
                   + X(SAN-DIEGO,TOPEKA) =L= 600 ; (LHS = 0)

---- DEMAND        =G=  satisfy demand at market j

DEMAND(NEW-YORK).. X(SEATTLE,NEW-YORK) + X(SAN-DIEGO,NEW-YORK) =G= 325 ;
                  (LHS = 0 ***)

DEMAND(CHICAGO).. X(SEATTLE,CHICAGO) + X(SAN-DIEGO,CHICAGO) =G= 300 ;
                  (LHS = 0 ***)

DEMAND(TOPEKA).. X(SEATTLE,TOPEKA) + X(SAN-DIEGO,TOPEKA) =G= 275 ;
                 (LHS = 0 ***)
```

方程式列表显示从在 GAMS 输入中指定的分区(block)生成的单独的约束.在 GAMS 中使用者可以以一种非常紧凑的形式写下被索引的方程式分区(block),这将产生大量的单个方程式.在我们的示例中,我们指定了 3 个方程式分区,生成了 6 个单独的方程式.

列列表

```
---- X          shipment quantities in cases

X(SEATTLE,NEW-YORK)
      (.LO, .L, .UP = 0, 0, +INF)
-0.225  COST
      1      SUPPLY(SEATTLE)
      1      DEMAND(NEW-YORK)

X(SEATTLE,CHICAGO)
      (.LO, .L, .UP = 0, 0, +INF)
-0.153  COST
      1      SUPPLY(SEATTLE)
      1      DEMAND(CHICAGO)

X(SEATTLE,TOPEKA)
      (.LO, .L, .UP = 0, 0, +INF)
-0.162  COST
      1      SUPPLY(SEATTLE)
      1      DEMAND(TOPEKA)

REMAINING 3 ENTRIES SKIPPED

---- Z          total transportation costs in thousands of dollars

Z
      (.LO, .L, .UP = -INF, 0, +INF)
      1      COST
```

列列表提供信息到生成的单独的变量上.变量 X(I,J)扩展出 6 个单独的变量.当许多变量从一个分区中被生成,默认列表只显示最初的 3 个(用户可以修改).

求解信息

MODEL STATISTICS

BLOCKS OF EQUATIONS	3	SINGLE EQUATIONS	6
BLOCKS OF VARIABLES	2	SINGLE VARIABLES	7
NON ZERO ELEMENTS	19		

GENERATION TIME = 0.017 SECONDS

EXECUTION TIME = 0.033 SECONDS VERID AXU-25-085

S O L V E S U M M A R Y

MODEL	TRANSPORT	OBJECTIVE	Z
TYPE	LP	DIRECTION	MINIMIZE
SOLVER	BDMLP	FROM LINE	47

**** SOLVER STATUS 1 NORMAL COMPLETION
**** MODEL STATUS 1 OPTIMAL
**** OBJECTIVE VALUE 153.6750

RESOURCE USAGE, LIMIT	0.184	1000.000
ITERATION COUNT, LIMIT	4	1000

B D M L P 1.1 --- AXP/OSF 1.1.045-017

A. Brooke, A. Drud, and A. Meeraus,
Analytic Support Unit,
Development Research Department,
World Bank,
Washington, D.C. 20433, U.S.A.

EXIT -- OPTIMAL SOLUTION FOUND.

求解声明将生成模型(单个方程式和对应到特定模型的变量的产物).首先一些关于生成的模型的统计表将会被显示:方程式数,变量和非零元素.

在求解汇总信息部分,我们看到 BDMLP 被调用来求解这个模型.BDMLP 经过 4 次反复,耗时 0.18 秒找到了这个问题的最优解.求解信息下列的消息来自求解器.

解(Solution)

```

                LOWER    LEVEL    UPPER    MARGINAL
---- EQU COST          .        .        .        1.000
    COST      define objective function

---- EQU SUPPLY      observe supply limit at plant i
                LOWER    LEVEL    UPPER    MARGINAL
SEATTLE        -INF    350.000  350.000    EPS
SAN-DIEGO      -INF    550.000  600.000    .

---- EQU DEMAND      satisfy demand at market j
                LOWER    LEVEL    UPPER    MARGINAL
NEW-YORK       325.000  325.000   +INF    0.225
CHICAGO        300.000  300.000   +INF    0.153
TOPEKA         275.000  275.000   +INF    0.126

---- VAR X           shipment quantities in cases
                LOWER    LEVEL    UPPER    MARGINAL
SEATTLE .NEW-YORK    .        50.000   +INF    .
SEATTLE .CHICAGO     .        300.000  +INF    .
SEATTLE .TOPEKA      .         .        +INF    0.036
SAN-DIEGO.NEW-YORK  .        275.000  +INF    .
SAN-DIEGO.CHICAGO   .         .        +INF    0.009
SAN-DIEGO.TOPEKA    .        275.000  +INF    .

                LOWER    LEVEL    UPPER    MARGINAL
---- VAR Z           -INF    153.675   +INF    .

    Z           total transportation costs in thousands of dollars

**** REPORT SUMMARY :           0      NONOPT
                                0      INFEASIBLE
                                0      UNBOUNDED

```

解被显示在这里.边际值(marginals)对应方程式的重复和变量减少的花费.

写工具不需要学习一门其它的语言.在 GAMS 所有的数据处理,模型定义和报告编写都是在一个单独的环境中完成的.

Dantzig G. B., Linear Programming and Extensions, Princeton University Press, Princeton, New Jersey, 1963, Chapter 3-3.

*通用代数建模系统(GAMS)是一套用于数学规划和最优化的高级建模系统.它包含了一个语言编译器和一套稳定的集成的高性能求解器.GAMS 是一套特别用来处理复杂的,大型的建模问题的应用程序,允许您建立大型的可维护的模型,使之能够快速的适应新的情况.



上海卡贝信息技术有限公司 - 专业统计/质量/数据分析软件代理商,致力于为高校及企业提供专业质量统计数据类分析软件.主要代理销售产品有:

- ◇ MINITAB 14 质量统计数据软件
- ◇ JMP 6 统计知识发现软件/数据分析/DOE 分析软件
- ◇ LISREL 8 结构方程分析建模软件
- ◇ EViews 5 计量经济学软件/时间序列分析
- ◇ RATS 6 计量经济学软件/时间序列分析
- ◇ OxMetrics 4 计量经济学分析, 预测和建模整体解决方案
- ◇ LIMDEP 8 计量经济学软件/时间序列分析
- ◇ HLM 分层线性模型分析软件
- ◇ @RISK 4.5 风险评估与决策分析软件
- ◇ PrecisionTree 风险评估与决策分析软件(决策树创建)
- ◇ Expert Choice AHP 专家决策分析软件
- ◇ LINGO 10 运筹学软件/模型优化
- ◇ GAMS 2.50 运筹规划分析软件

更多信息欢迎访问 <http://www.cabit.com.cn>

电话: 021-50391085 50391087

Email: sales@cabit.com.cn